274 CS 538 Spring 2002^©

Function Types Can be Polytypes

Recall that'a, 'b, ... represent type variables. That is, any valid type may be substituted for them when checking type correctness. ML said the type ofappend is

val append = fn :
 'a list * 'a list -> 'a list

Why does'a appear in three places? We can defineeitherNull, a predicate that determines whether either of two lists is null as

fun eitherNull(L1,L2) =
 null(L1) orelse null(L2);
val eitherNull =
 fn : ’a list * ’b list -> bool

Why are both'a and 'b used in

eitherNull’s type?

275 CS 538 Spring 2002^©

Currying

ML chooses the most general (leastrestrictive) type possible for userdefined functions. Functionsarefirst-classobjects, asin Scheme. The function definition

fun f x y = expression;

defines a functionf (of x) that returns a function (ofy). Reducing multiple argument functions to a sequence of one argument functions is calledcurrying (afterHaskellCurry, a mathematician who popularized the approach).

276 CS 538 Spring 2002^©

Thus

fun f x y = x :: [y];
val f = fn : 'a -> 'a -> 'a list

says thatf takes a parameterx, of type'a, andreturnsa function(ofy, whose type is'a) that returns a list of 'a. Contrast this with the more conventional

fun g(x,y) = x :: [y];
val g = fn : 'a * 'a -> 'a list

Hereg takes a pair of arguments (eachof type'a) andreturnsa value of type'a list. The advantage of currying is that we canbindoneargumentandleavethe remaining argument(s) free.

277 CS 538 Spring 2002^©

For example

f(1);

isa legalcall.It returnsa functionof type

fn : int -> int list

Thefunctionreturnedisequivalentto

fun h b = 1 :: [b];
val h = fn : int -> int list

278 CS 538 Spring 2002^©

Map Revisited

MLsupportsthe map function,which can be defined as

fun map(f,[]) = []
  | map(f,x::y) =
      (f x) :: map(f,y);

val map =
 fn : ('a -> 'b) * 'a list -> 'b list

Thistypesaysthat map takesa pairof arguments. One is a function from type'a to type'b. The second argument is a list of type'a. The result is a list of type'b. In curried formmap is defined as

fun map f [] = []
  | map f (x::y) =
      (f x) :: map f y;

val map =
 fn : ('a -> 'b) ->
       'a list -> 'b list

279 CS 538 Spring 2002^©

This type says thatmap takes one argumentthat isafunctionfromtype 'a to type'b. It returns a function that takes an argument that is a list of type'a and returns a list of type

'b.

The advantage of the curried form of map is that we can now usemap to create “specialized” functions in which the function that is mapped is fixed. For example,

val neg = map not;
val neg =
  fn : bool list -> bool list
neg [true,false,true];
val it = [false,true,false] :
 bool list

280 CS 538 Spring 2002^©

Power Sets Revisited

Let’s compute power sets in ML. We want a functionpow that takes a list of values, viewed as a set, and which returns a list of lists. Each sublist will be one of the possible subsets of the original argument. For example,

pow [1,2] =  [[1,2],[1],[2],[]]

We first define a version ofcons in curried form:

fun cons h t = h::t;
val cons = fn :
  'a -> 'a list -> 'a list

281 CS 538 Spring 2002^©

Now we definepow. We define the powerset of the empty list,[], to be

[[]]. That is, the power set of the

empty set is set that contains only the empty set. For a non-empty list, consisting of

h::t,wecomputethepowersetoft,

which we callpset. Then the power set forh::t is justh distributed throughpset appended topset. We distributeh throughpset very elegantly: we just map the function

(cons h)to pset. (cons h) addsh

to the head of any list it is given. Thus mapping(cons h) to pset addsh to all lists inpset.

282 CS 538 Spring 2002^©

The complete definition is simply

 fun pow [] = [[]]
 |  pow (h::t) =
    let
     val pset = pow t
    in
    (map (cons h) pset) @ pset
    end;
val pow =
  fn : 'a list -> 'a list list

Let’s trace the computation of

pow [1,2].

Here h = 1 andt = [2]. We needto computepow [2]. Nowh = 2 and t = []. We knowpow [] = [[]],

sopow [2] =

(map (cons 2) [[]])@[[]] =
([[2]])@[[]] = [[2],[]]

283 CS 538 Spring 2002^©

Thereforepow [1,2] =

(map (cons 1) [[2],[]])
 @[[2],[]] =
[[1,2],[1]]@[[2],[]] =
[[1,2],[1],[2],[]]

284 CS 538 Spring 2002^©

Composing Functions

Wecandefineacompositionfunction that composes two functions into one:

fun comp (f,g)(x) = f(g(x));
val comp = fn :
('a -> 'b) * ('c -> 'a) ->
 'c -> 'b

In curried form we have

fun comp f g x = f(g(x));
val comp = fn :
('a -> 'b) ->
('c -> 'a) -> 'c -> 'b

For example,

fun sqr x:int = x*x;
val sqr = fn : int -> int
comp sqr sqr;
val it = fn : int -> int

285 CS 538 Spring 2002^©

comp sqr sqr 3;
val it = 81 : int

In SMPo (lower-case O) is the infix composition operator. Hence

sqr o sqr ≡ comp sqr sqr

286 CS 538 Spring 2002^©

Lambda Terms

ML needs a notation to write down unnamed (anonymous) functions, similar to the lambda expressions Scheme uses. That notation is

fn arg => body;

For example,

val sqr = fn x:int => x*x;
val sqr = fn : int -> int

In fact the notation used to define functions,

fun name arg = body;

isactuallyjustanabbreviationforthe more verbose

val name = fn arg => body;

287 CS 538 Spring 2002^©

An anonymous function can be used wherever a function value is needed. For example,

map (fn x => [x]) [1,2,3];
val it =
[[1],[2],[3]] : int list list

We can use patterns too:

(fn [] => []
  |(h::t) => h::h::t);
val it = fn : 'a list -> 'a list

(What does this function do?)

288 CS 538 Spring 2002^©

Polymorphism vs. Overloading

ML supports polymorphism. A function may accept a polytype (a set of types) rather than a single fixed type. In all cases, the same function definition is used. Details of the supplied type are irrelevant and may be ignored. For example,

fun id x = x;
val id = fn : 'a -> 'a
fun toList x = [x];
val toList = fn : 'a -> 'a list

289 CS 538 Spring 2002^©

Overloading, as in C++ and Java, allows alternative definitions of the same method or operator, with selection based on type. Thus in Java+ may represent integer addition, floating point addition or string concatenation, even though these are really rather different operations. In ML+, -, * and = are overloaded. When= isused(to test equality),ML deduces that anequality type is required.(Most,butnotall,typescan be compared for equality). When ML decides an equality type is needed, it uses a type variable that begins with two tics rather than one.

fun eq(x,y) = (x=y);
val eq = fn : ''a * ''a -> bool

290 CS 538 Spring 2002^©

Defining New Types in ML

Wecancreatenewnamesforexisting types (type abbreviations) using

type id = def;

For example,

type triple = int*real*string;
type triple = int * real * string
type rec1=
 {a:int,b:real,c:string};
type rec1 =
  {a:int, b:real, c:string}
type 'a triple3  = 'a*'a*'a;
type 'a triple3 = 'a * 'a * 'a
type intTriple = int triple3;
type intTriple = int triple3

Thesetypedefinitionsareessentiality macro-like name substitutions.

291 CS 538 Spring 2002^©

The Datatype Mechanism

New types are defined using the

datatype mechanism, which

specifies new datavalue constructors. For example,

datatype color =
 red|blue|green;
datatype color =
 blue | green | red

Pattern matching works on userdefined types using their constructors:

fun translate red = "rot"
  | translate blue = "blau"
  | translate green = "gruen";
val translate =
 fn : color -> string

292 CS 538 Spring 2002^©

fun jumble red = blue
  | jumble blue = green
  | jumble green = red;
val jumble = fn : color -> color
translate (jumble green);
val it = "rot" : string

SML Examples

Source code for most of the SML examples presented here may be found in

~cs538-1/public/sml/class.sml

293 CS 538 Spring 2002^©

Parameterized Constructors

The constructors used to define data types may be parameterized:

datatype  money =
   none
 | coin of int
 | bill of int
 | iou of real * string;
datatype money =
   bill of int | coin of int
 | iou of real * string | none

Now expressions likecoin(25) or

bill(5) or iou(10.25,"Andy")

represent valid values of typemoney.

294 CS 538 Spring 2002^©

We can also define values and functions of typemoney:

val dime = coin(10);
val dime = coin 10 : money
val deadbeat =
iou(25.00,"Homer Simpson");
val deadbeat =
 iou (25.0,"Homer Simpson") :
 money
fun amount(none) = 0.0
 |  amount(coin(cents)) =
     real(cents)/100.0
 |  amount(bill(dollars)) =
     real(dollars)
 |  amount(iou(amt,_)) =
     0.5*amt;
 val amount = fn : money -> real